

## APPENDIX IV-D

### STJ-EROS ARC MACRO LANGUAGE (AML) CODE

```

/*****
/* APPENDIX IV-D
/*
/* FILE NAME:   stj_eros.aml
/*
/* PROGRAMMER:  Carlos E. Ramos Scharron
/*              Colorado State University- Ft Collins CO
/*
/* DATE:        November 25, 2003
/*
/* PURPOSE:     This program, written in Arc Macro Language is intended to
/*              estimate rates of erosion and sediment delivery for unpaved
/*              roads and natural processes on the island of St. John,
/*              US Virgin Islands.
/*
/* RUN COMMAND: Once in the correct workspace type '&run stj_eros' at the ARC
/*              command line.
/*
/* DEPENDENCIES: This program is run in ArcInfo 8.2 and in its Tables module.
/*
/* TERMINATING  This program does not call any additional programs once it
/* CONVENTIONS: terminates.
/*
/* CALLING      The program requires the user to define the following
/* CONVENTIONS: variables:
/*              (1) %BASIN% as the name of the basin (watershed) for which
/*              analysis is to be conducted.
/*              (2) %YEARS% as the total number of years over which erosion
/*              and sediment delivery are to be calculated
/*              (3) %RAIN_RATE% as an annual rainfall rate.
/*              (4) %ROAD_NAME% as the name for the output file, data layer,
/*              and shapefile containing road erosion and sediment delivery
/*              results.
/*              (5) %NAT_NAME% as the name for output file, data layer, and
/*              and shapefile containing natural erosion and sediment
/*              delivery results.
/*
/* TERMINATING  This program does not call any additional programs when it
/* CONVENTIONS: terminates.
/*
/* ALGORITHMS:  After initializing variables that will be used to calculate
/*              sediment production rates, it performs some housekeeping
/*              commands to delete any pre-existing GIS data layers with
/*              the same names as those that will be created by the
/*              program. An interface then asks the user whether to
/*              continue with a brief on-screen introduction to the
/*              program or to move directly to begin entering model
/*              parameters. The MODEL routine is then invoked to control
/*              the general flow of the model. Once all of the routines are
/*              called the program asks the user whether to view a summary
/*              of the results by invoking the SUMMARY_RESULTS routine or
/*              to exit the program. Before exiting an on-screen message
/*              reminds the user of the names given to the GIS data layer
/*              and text files containing the model results.
/*
/*****

&terminal 9999

&sv treethrow = 0.17 /* tons per km of channel length per year
&sv bank_er = 10 /* kg per m^2 of bank surface per year
&sv undisturbed = 0.000064 /* kg per m^2 of undisturbed hillslope surface
/* per cm of precipitation occurring during events exceeding 6 cm of rain
&sv abandoned = 0.071 /* kg per m^2 of road surface, cm of precipitation
/* unit gradient
&sv graded = 4.73 /* kg per m^2 of road surface, cm of precipitation &
/* unit gradient to the 1.5th power
&sv ungraded = 1.88 /* kg per m^2 of road surface, cm of precipitation &
/*unit gradient to the 1.5th power
&sv cutslope = 9 /* % of sediment produced at the segment scale

```

```
&sv un_sus_fr = 0.59 /* Fraction of sediment produced from ungraded roads
/* that is finer than 2 mm
&sv gr_sus_fr = 0.64 /* Fraction of sediment produced from graded roads
/* that is finer than 2 mm
&sv ab_sus_fr = 0.40 /* Fraction of sediment produced from abandoned
/* roads that is finer than 2mm
&sv bank_sus_fr = 0.75 /* Fraction of sediment produced from streambanks
/* that is finer than 2mm
&sv tree_sus_fr = 0.75 /* Fraction of sediment produced from treethrow
/* that is finer than 2mm
&sv se_sus_fr = 0.577 /* Fraction of sediment produced from undisturbed
/* hillslopes that is finer than 2 mm
&sv silt_u_rd_fr = 0.04 /* Fraction of sediment produced from ungraded
/* roads that is in the silt-size category
&sv silt_g_rd_fr = 0.06 /* Fraction of sediment produced from graded
/* roads that is in the silt-size category
&sv silt_a_rd_fr = 0.001 /* Fraction of sediment produced from abandoned
/* roads that is in the silt-size category
&sv silt_se_fr = 0.004 /* Fraction of sediment produced from undisturbed
/* hillslopes that is in the silt-size category
&sv silt_loss = 9 /* Silt and clay estimated loss ratio of true yield
/* to that measured by sediment traps
&if [exists rds_del -cover] &then kill rds_del all
&if [exists drain -cover] &then kill drain all
&if [exists dra_del -cover] &then kill dra_del all
&if [exists del_bd -cover] &then kill del_bd all
&if [exists temp_rd -cover] &then kill temp_rd all
&if [exists temp_del -cover] &then kill temp_del all
&if [exists pt_del -cover] &then kill pt_del all
&if [exists bank_del -cover] &then kill bank_del all
&if [exists str_del -cover] &then kill str_del all
&if [exists str_bd -cover] &then kill str_bd all
&if [exists buf_str -cover] &then kill buf_str all
&if [exists temp_tre -cover] &then kill temp_tre all
&if [exists tr_del -cover] &then kill tr_del all
&if [exists se_bd -cover] &then kill se_bd all
&if [exists temp_bd -cover] &then kill temp_bd all
&if [exists buf_cst -cover] &then kill buf_cst all
&if [exists cst_bd -cover] &then kill cst_bd all


&type ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
&type ^                                                                                   ^
&type ^      Welcome to the St. John Erosion and Sediment Delivery ArcGIS Model!       ^
&type ^                                                                                   ^
&type ^              Author: Carlos E. Ramos Scharron                                ^
&type ^           Colorado State University, Fort Collins CO                          ^
&type ^                                                                                   ^
&type ^                      November 25, 2003                                         ^
&type ^                                                                                   ^
&type ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^

&if [query 'Press enter for brief introduction or "no" to start entering model ~
parameters.' .true.] &then &call intro
&call model
&if [query 'Press enter for brief summary of results or "no" to exit.' .true.] ~
    &then &call summary_results
&type /&^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
&type ^
&type ^ Road erosion and sediment delivery results are available at:
&type ^      %road_name%.txt and %road_name% line data layer.
&type ^ The order of column (items) shown in %road_name%.txt
&type ^ is as follows:
&type ^     length_m surface grading potential rd_pr_rate
&type ^     rd_ttl_del_rate rd_sus_del_rate cs_pr_rate cs_ttl_del_rate
&type ^     cs_sus_del_rate
&type ^
&type ^ Natural erosion and sediment delivery results are available at:
&type ^      %nat_name%.txt and %nat_name% polygon data layer.
&type ^ The order of column (items) shown in %nat_name%.txt
&type ^ is as follows:
```

```

&type ^      area potential se_ton se_ttl_del se_sus_del bank_ton
&type ^      bank_ttl_del bank_sus_del tree_ton tree_ttl_del tree_sus_del
&type ^      nat_pr_rate nat_t_del_rate nat_s_del_rate
&type ^
&type ^ The user may find the following covers useful for display purposes:
&type ^      %basin%_bd      del_bd      pt_del
&type ^
&type ^ ~~~~~~
&return

```

```

&ROUTINE MODEL
/*****
/*
/* PURPOSE:      Controls the flow of the model by calling routines.
/*
/* REQUIRED      None
/* PRE-DEFINED
/* VARIABLES:
/*
/* IMPORTANT    None
/* CREATED
/* VARIABLES:
/*
/* SIDE         None
/* EFFECTS:
/*
/* ALGORITHM:  Routines are called in the following order:
/*
/*      (1) SET_SDR- Assigns the value of variables %HI_POT% and
/*      %MOD_POT% according to values chosen from a menu or entered
/*      by the user. These two variables are used as sediment delivery
/*      ratios (SDR's) to estimate sediment yields.
/*      (2) DEL_POTENTIAL- Assigns sediment delivery ratio values to the
/*      entire basin of interest according to its pre-determined
/*      sediment delivery potential.
/*      (3) SET_YEARS- Assigns the value of variable %YEARS% according
/*      to a value chosen from a menu or entered by the user. This
/*      variable defines the total number of years over which sediment
/*      yields are to be estimated.
/*      (4) SET_RAIN- Assigns the value of variable %RAIN_RATE%
/*      according to a value chosen from a menu or entered by the user.
/*      %RAIN_RATE% defines the annual rainfall rate in cm per year.
/*      (5) ROADS_NAME- Assigns the variable %ROAD_NAME% according to
/*      a value chosen from a menu or entered by the user. %ROAD_NAME%
/*      is the name that will be given to the GIS data layer and text
/*      file containing the sediment yield estimates for all road-
/*      related sediment sources.
/*      (6) NAT_NAME- Assigns the value of variable %NAT_NAME% according
/*      to a value chosen from a menu or entered by the user.
/*      %NAT_NAME% is the name of the GIS data layer and text file
/*      containing the sediment yield estimates for all natural sediment
/*      sources.
/*
/*      Once all parameters have been chosen by the user, the program
/*      provides an on-screen list of all of the parameters
/*      that were chosen. The user is ask whether to exit the program
/*      to re-enter the parameter values or to run the program with
/*      the values already chosen. Exiting quits the program, otherwise
/*      the program calls routine #7.
/*
/*      (7) RD_EROSION- Calculates sediment production and yield rates
/*      for all road-related sediment sources and stores them into a
/*      line GIS data layer and into a text file.
/*      (8) STREAMBANK- Calculates streambank sediment production and
/*      yield rates and stores them into a polygon GIS data layer.
/*      (9) STREAM_TOTAL- Calculates treethrow sediment production and
/*      yield rates and stores them into a polygon GIS data layer. The
/*      final layer also contains streambank sediment yield estimates.
/*      (10) SURF_EROSION- Calculates sediment production and yield
/*      rates from undisturbed surfaces and stores them into a polygon

```

```

/*          GIS data layer.          */
/*          */
/*          (11) NAT_EROSION- Combines streambank, treethrow, and surface */
/*          erosion and yield estimates into a single polygon GIS data */
/*          layer.          */
/*          */
/*****/

&call set_SDR
&call del_potential
&call set_years
&call set_rain
&call roads_name
&call nat_name
&type +*****+
&type +
&type + The following is a list of all user defined parameters:
&type +
&type + Basin: %watshed%
&type + Time: %years% year(s)
&type + Rainfall rate: %rain_rate% cm per year
&type + Total rainfall: %rain% cm
&type + Sediment delivery ratio for high potential areas: %hi_pot%
&type + Sediment delivery ratio for moderate potential areas: %mod_pot%
&type + Road erosion and sediment delivery results will be available at:
&type +      %road_name%.txt and %road_name% line data layer
&type + Natural erosion and sediment delivery results will be available at:
&type +      %nat_name%.txt and %nat_name% polygon data layer
&type +
&type +*****+&

&if [query 'Press enter to run model with these values or "no" to quit and ~
start over.' .true.] &then
    &do
        &call rd_erosion
        &call streambank
        &call stream_total
        &call surf_erosion
        &call nat_erosion
        &type /&All routines successfully completed!/&
    &end
&else
    &do
        &type /&Bailing out.../&
        &return
    &end
&return

&ROUTINE INTRO

&type /&*****+
&type /&This model is intended to estimate the rates of sediment production and
&type delivery into the marine environment in St. John, USVI.
&type /&Rates of sediment production are estimated from unpaved road surfaces,
&type cutslopes as well as from natural sources. Any changes in the default
&type values used in this model can be made by making the appropriate changes
&type in the AML file only when they are supported by scientifically-sound
&type evidence.
&type /&The regression equations and erosion rates currently set for this
&type model are:
&type /& Treethrow sediment production rates:
&type      %treethrow% kg per m of stream per year
&type Streambank sediment production rates:
&type      %bank_er% kg per m^2 of bank area per year
&type Surface erosion on natural hillslopes:
&type      %undisturbed% kg per m^2 of hillslope area per cm of
&type      precipitation during events > 6 cm in total rainfall
&type Graded road erosion in kg per m^2 =
&type      (-0.432+(4.73 * slope^1.5 * cm of rainfall)
&type Ungraded road erosion in kg per m^2 =

```



```

&do
  &type /&ERROR!!! SDR for high potential areas must be greater than 0.5!
  &call set_sdr
&end
&else
&do
  &if %hi_pot% > 1 &then
    &do
      &type /&ERROR!!! SDR for high potential areas cannot be greater than 1.0!
      &call set_sdr
    &end
  &end
&if %mod_pot% < 0.0 &then
  &do
    &type /&ERROR!!! SDR for moderate potential areas cannot be negative!
    &call set_sdr
  &end
&else
&do
  &if %mod_pot% > 0.5 &then
    &do
      &type /&ERROR!!! SDR for moderate potential aras cannot be greater than 0.5!
      &call set_sdr
    &end
  &end
&return

```

# &ROUTINE DEL\_POTENTIAL

```

/*****
/*
/* PURPOSE:    Assigns sediment delivery ratio values to the entire basin of
/*             interest according to its pre-determined sediment delivery
/*             potential.
/*
/* REQUIRED     HI_POT is the sediment delivery ratio (SDR) assigned to areas
/* PRE-DEFINED which have been identified as having a high potential for sediment
/* VARIABLES:  delivery to the marine environment.
/*             MOD_POT is the SDR for areas with a moderate delivery potential.
/*
/* IMPORTANT   %BASIN% is the code name of the basin for which analysis is to be
/* CREATED     conducted. The user chooses the name from a menu listing the
/* VARIABLES:  currently available choices.
/*             %WATSHED% is the full name of the basin chosen for analysis.
/*
/* SIDE        (1)A new polygon cover called del_bd is created.
/* EFFECTS     (2)TABLES is invoked.
/*
/* ALGORITHM:  The user is prompted to choose a basin name from a menu list. The
/*             choice is assigned as the value of variable %BASIN%. The value of
/*             variable %WATSHED% is assigned according to the value of %BASIN%.
/*             In TABLES a new item called DEL_RATIO is added to the table.
/*             a new cover is created by clipping the sed_del layer with the
/*             %basin%_bd layer. Sediment delivery ratios chosen by the
/*             user in the SET_SDR routine are then assigned to this new
/*             layer. The numerical value of DEL_RATIO for each polygon is
/*             assigned according to its POTENTIAL and to the pre-established
/*             values of the variables %HI_POT% and %MOD_POT%.
/*
/* *****/

```

```

&type /&The user is now required to choose a basin for analysis. The model can be
&type applied to the following basins:

```

```

&type /& Cinnamon Bay (cb)
&type   Fish Bay (fb)
&type   Lameshur Bay (lb)/&
&if [query 'Press enter to continue.' .true.] &then &type Continuing...
&sv basin = [ getchoic cb fb lb -prompt 'Select a basin for analysis' ]
&select %basin%
  &when cb
    &sv watshed = Cinnamon Bay

```

```

&when fb
  &sv watshed = Fish Bay
&when lb
  &sv watshed = Lameshur Bay
&end
&type /&Creating sediment delivery cover for %watshed% .../&
clip sed_del %basin%_bd del_bd poly
tables
select del_bd.pat
additem del_bd.pat del_ratio 8 8 n 1
reselect potential = 'high'
calculate del_ratio = %hi_pot%
aselect
reselect potential = 'moderate'
calculate del_ratio = %mod_pot%
aselect
reselect potential = 'wetland'
calculate del_ratio = %mod_pot%
reselect potential = 'no'
calculate del_ratio = 0
quit
&type Done creating sediment delivery cover for %watshed%...
&return

```

```

&ROUTINE SET_YEARS
/*****
/*
/* PURPOSE:      Assigns the value of variable %YEARS% according to a value
/*               chosen from a menu or entered by the user.
/*
/*
/* REQUIRED      None
/* PRE-DEFINED
/* VARIABLES:
/*
/* IMPORTANT    %YEARS% is the total number of years for which sediment production
/* CREATED      and delivery are to be estimated.
/* VARIABLES:
/*
/* SIDE         None
/* EFFECTS:
/*
/* ALGORITHM:   The user is prompted to choose the number of years over which
/*               sediment production and delivery are to be estimated from a number
/*               of options. The user is given the option to enter another value not
/*               offered in the menu. The value entered is then rounded and
/*               assigned to the %YEARS% variable. The years value is then tested
/*               to verify that it is greater than 0 and does not exceed 50. If this
/*               is not satisfied an error message prints on the screen and the user
/*               is asked to once again enter a value.
/*
*****/

```

```

&type /& Now the user will be requested to enter the total number of years
&type over which sediment production and sediment delivery will be calculated.
&type The number of years can be chosen by selecting a value from choices in
&type a menu or by entering a value once the "other" option has been selected.
&type Please note that any real values entered manually will be rounded and
&type that the model will only accept values between 1 and 50 years./&
&if [query 'Press enter to continue.' .true.] &then &type Continuing...
&sv time [getchoice 1 5 10 25 50 -prompt 'Choose total number of years' -other]
&sv years = [round %time%]
&if %years% <= 0 &then
  &do
    &type /&ERROR!!! Number of years must be greater than 0!
    &call set_years
  &end
&else
  &do
    &if %years% > 50 &then

```



```

        &do
            &type /&ERROR!!! Number of years must be less than or equal to 50.
            &call set_years
        &end
    &end
&return

&ROUTINE SET_RAIN
/*****
/*
/* PURPOSE:      Assigns the value of variable %RAIN_RATE% according to a value
/*               chosen from a menu or entered by the user.
/*
/* REQUIRED      %YEARS% is the total number of years chosen by the user in the
/* PRE-DEFINED  SET_YEARS routine.
/* VARIABLES:
/*
/* IMPORTANT    %RAIN_RATE% is the rainfall (in cm per year) for which sediment
/* CREATED      production and delivery are to be estimated.
/* VARIABLES:    %RAIN% is the total rainfall (in cm) used to calculate sediment
/*               production from roads and undisturbed hillslopes.
/*
/* SIDE         None
/* EFFECTS:
/*
/* ALGORITHM:    The user is prompted to choose an annual rainfall rate for which
/*               sediment production and delivery are to be estimated from a number
/*               of options. The user is given the option to enter another value not
/*               offered in the menu. The value entered is then rounded and assigned
/*               to the %RAIN_RATE% variable. The %RAIN_RATE% value is then tested to
/*               verify that it is greater than 70 and does not exceed 160. If this
/*               test is not satisfied an error message prints on the screen and the
/*               user is asked to enter another value. The %RAIN% value is then
/*               calculated as the product of %RAIN_RATE% times %YEARS%.
/*
*****/

&type /& Now the user will be requested to enter an annual rainfall rate in
&type cm per year. A rate can be chosen by selecting a value from choices in
&type a menu or by entering a value once the "other" option has been selected.
&type Please note that any real values entered manually will be rounded and
&type that the model will only accept values between 70 and 160 cm per year./&
&type Mean annual rainfall in St. John is spatially variable. While the
&type driest portions on East End and lower elevations of Fish Bay and
&type Lameshur Bay basins receive approximately 90 cm of rainfall per year,
&type the highest elevations around Bordeaux Mountain typically receive
&type about 130 cm per year. A 115 cm per year value is considered to be a
&type reasonable approximation to long-term mean rainfall rates for most
&type basins on the island. /&
&if [query 'Press enter to continue.' .true.] &then &type Continuing...
&sv annual_rain [getchoice 90 115 140 -prompt 'Choose an annual rainfall rate' -other]
&sv rain_rate = [round %annual_rain%]
&if %rain_rate% < 70 &then
    &do
        &type /&ERROR!!! Rainfall rate must be greater than 70 cm per year!
        &call set_rain
    &end
&else
    &do
        &if %rain_rate% > 160 &then
            &do
                &type /&ERROR!!! Rainfall rate must be less than or equal to 160 cm per year.
                &call set_rain
            &end
        &end
    &end
&sv rain = [calc %rain_rate% * %years%]
&return

```

```

&ROUTINE ROADS_NAME
/*****
/*
/* PURPOSE:    Assigns the value of variable %ROAD_NAME% according to a value
/*             chosen from a menu or entered by the user. %ROAD_NAME% is the
/*             name of the new output coverage and text file to be created by
/*             this model.
/*
/*
/* REQUIRED    %BASIN% is the code name of the basin for which analysis is to be
/* PRE-DEFINED conducted.
/* VARIABLES:
/*
/* IMPORTANT  %ROAD_NAME% is the user chosen name used for the layer and text
/* CREATED    file containing the results of the sediment production and delivery
/* VARIABLES: estimates for roads.
/*             %NAME_LENGTH% is the number of characters used by %ROAD_NAME%.
/*
/*
/* SIDE       None
/* EFFECTS:
/*
/*
/* ALGORITHM: The user is prompted to choose a name from a number of options
/*             in a menu. The user is given the option to enter another name not
/*             offered in the menu. The number of characters in the chosen name
/*             is assigned to variable %NAME_LENGTH% which is used to test whether
/*             the chosen name has been given a name less than 8 characters in
/*             length. If this test is not satisfied an error message prints on
/*             the screen and the user is asked to enter another name.
/*
/*
*****/

```

```

&type /& Now the user will be requested to enter the names of the final coverages
&type and txt files containing the sediment production and delivery estimates
&type for roads and natural erosion processes. The names can be chosen by
&type selecting a value from choices in a menu or by entering a value once the
&type "other" option has been selected. Please note that the model will only take
&type names that are less than or equal to 8 characters in length./&
&if [query 'Press enter to continue.' .true.] &then &type Continuing...
&sv road_name [getchoice %basin%_rd r_%basin% -prompt~
'Choose a name for road erosion layer and results file' -other]
&sv name_length = [length %road_name%]
&if %name_length% = 0 &then
&do
&type /&ERROR!!! Please enter a name!
&call roads_name
&end
&if %name_length% > 8 &then
&do
&type /&ERROR!!! Please, enter a name with no more than 8 characters!
&call roads_name
&end
&if [exists %road_name% -cover] &then
&do
&type /&Cover %road_name% already exists!!
&if [query 'Delete current cover (yes or no)' .false.]~
&then kill %road_name% all
&else &call roads_name
&end
&return

```

```

&ROUTINE NAT_NAME
/*****
/*
/* PURPOSE:    Assigns the value of variable %NAT_NAME% according to a value
/*             chosen from a menu or entered by the user. %NAT_NAME% is the
/*             name of the new natural erosion coverage and text file to
/*             be created by this model.
/*
/*
/* REQUIRED    %BASIN% is the code name of the basin for which analysis is to be
/* PRE-DEFINED conducted.
/* VARIABLES:

```

```

/*
/* IMPORTANT      %NAT_NAME% is the user chosen name used for the cover and text
/* CREATED        file containing the results of the sediment production and
/* VARIABLES:     delivery estimates for natural processes.
/*               %NAME_LENGTH% is the number of characters used by %NAT_NAME%.
/*
/* SIDE           None
/* EFFECTS:
/*
/* ALGORITHM:     The user is prompted to choose a name from a number of options
/*               in a menu. The user is given the option to enter another name not
/*               offered in the menu. The number of characters in the chosen name
/*               is assigned to variable %NAME_LENGTH% which is used to test whether
/*               the chosen name has been given a name less than 8 characters in
/*               length. If this test is not satisfied an error message prints on
/*               the screen and the user is asked to enter another name.
/*
/*****
&sv nat_name [getchoice %basin%_nat n_%basin% -prompt~
'Choose a name for natural erosion layer and results file' -other]
&sv name_length = [length %nat_name%]
&if %name_length% = 0 &then
&do
&type /&ERROR!!! Please enter a name!
&call nat_name
&end
&if %name_length% > 8 &then
&do
&type /&ERROR!!! Please, enter a name with no more than 8 characters!
&call nat_name
&end
&if %nat_name% = %road_name% &then
&do
&type /&ERROR!!! Please, enter a name other than %road_name%!
&call nat_name
&end
&if [exists %nat_name% -cover] &then
&do
&type Cover %nat_name% already exists!!!
&if [query 'Delete current cover (yes or no)' .false.]~
&then kill %nat_name% all
&else &call nat_name
&end
&return

&ROUTINE RD_EROSION
/*****
/*
/* PURPOSE:       Calculates sediment production and delivery rates for the basin
/*               chosen in the DEL_POTENTIAL routine using the attributes stored in
/*               GPS_RDS and GPS_DRA covers and the total rainfall and total time
/*               values chosen by the user during the SET_YEARS and SET_RAIN
/*               routines.
/*
/* REQUIRED        %BASIN% is the code name of the basin for which analysis is to be
/* PRE-DEFINED    conducted.
/* VARIABLES:     %UNGRADED% is the regression coefficient used to calculate
/*               sediment production from ungraded roads.
/*               %GRADED% is the regression coefficient used to calculate
/*               sediment production from graded roads.
/*               %ABANDONED% is the rate of sediment production from abandoned
/*               roads.
/*               %RAIN% is the total rainfall (in cm) used to calculate road
/*               sediment production rates.
/*               %YEARS% is the total time in years over which sediment production
/*               and delivery rates are calculated.
/*               %ROAD_NAME% is the user-given name of the final cover and text
/*               file that contain the final results of the road erosion and
/*               sediment delivery analysis.

```

```

/*          %SILT_LOSS% is the ratio of actual sediment production for the          */
/*          silt-size sediment fraction to that measured from sediment traps.      */
/*          %SILT_U_RD_FR%, %SILT_G_RD_FR%, %SILT_A_RD_FR%, %SILT_A_RD_FR% are    */
/*          the percent of sediment produced in the silt-size category for        */
/*          ungraded roads, graded roads, and abandoned roads, respectively.     */
/*          %UN_SUS_FR%, %GR_SUS_FR%, %AB_SUS_FR% are the percentages of          */
/*          sediment finer than 2 mm for ungraded roads, graded roads, and        */
/*          abandoned roads, respectively.                                         */
/*                                                                                */
/* IMPORTANT      None                                                            */
/* CREATED                                               */
/* VARIABLES:                                           */
/*                                                                                */
/* SIDE          (1)A new polygon cover called DRAIN is created and killed.        */
/* EFFECTS       (2)A new point cover called PT_DEL is created.                  */
/*              (3)A new line cover called %ROAD_NAME% is created.                */
/*              (4)TABLES is invoked.                                              */
/*              (5)A file called %ROAD_NAME%.txt is created.                      */
/*                                                                                */
/* ALGORITHM: The GPS_DRA point cover is clipped with the %BASIN%_BD cover to     */
/* eliminate all road drainage point outside of %BASIN%. The resulting          */
/* cover (DRAIN) is then intersected with the DEL_BD cover created                */
/* during the DEL_POTENTIAL routine to assign DEL_RATIO values to each          */
/* drainage structure according to its geographical location. The                */
/* %ROAD_NAME% line cover is copied from the GPS_RDS cover. TABLES is          */
/* invoked to dropitems from the PT_DEL.PAT and %ROAD_NAME%.AAT. A              */
/* joinitem function is invoked from ARC using the DRAIN_ID item                  */
/* as the relate item to assign DEL_RATIO values to %ROAD_NAME%.AAT.            */
/* TABLES is invoked once again to add and calculate the value of              */
/* items that will contain the sediment production and delivery                 */
/* estimates for each road segment within %BASIN% according to their            */
/* values stored in the following items: LENGTH_M, WIDTH_M, SLOPE,              */
/* GRADING, and DEL_RATIO. The variables %YEARS% and %RAIN% are also            */
/* used to calculate these items. Sediment delivery rates are                  */
/* estimated for both total loads and the fraction expected to be              */
/* carried as suspended load. Finally, the results are unloaded                 */
/* into a file called %ROAD_NAME%.txt before exiting TABLES.                  */
/*                                                                                */
/*****

&type /&Beginning rd_erosion routine...
&type /&Selecting drainage points within %watershed%...
clip gps_dra %basin%_bd drain point /* Eliminates all drainage points outside
/* of %basin%
intersect drain del_bd pt_del point
kill drain all
copy gps_rds %road_name%
tables
dropitem pt_del.pat (! drain# drain-id type del_bd# del_bd-id !)
dropitem %road_name%.aat source
quit
joinitem %road_name%.aat pt_del.pat %road_name%.aat drain_id
build %road_name% line
tables
additem %road_name%.aat temp_sed 12 12 n 1 /* Total road sed production
additem %road_name%.aat cslope_kg 10 10 n 1 /* Cutslope sediment production
/* equals 9% of total road sediment production
additem %road_name%.aat sed_kg 12 12 n 1 /* Road travelway sediment production
/* equals 91% of total road sediment production
additem %road_name%.aat rd_ton 8 8 n 2 /* Road travelway sed prod in tons
additem %road_name%.aat rd_del 8 8 n 2 /* Road travelway sed delivery in tons
additem %road_name%.aat cs_ton 8 8 n 1 /* Cutslope sed prod in tons
additem %road_name%.aat cs_del 8 8 n 2 /* Cutslope sed delivery in tons
additem %road_name%.aat rd_pr_rate 12 12 n 3 /* Travelway production rate
/* equals total production in tons divided by variable %YEARS%
additem %road_name%.aat rd_t_del_r 8 8 n 3 /* Travelway delivery rate
/* equals total delivery in tons divided by variable %YEARS%
additem %road_name%.aat cs_pr_rate 12 12 n 3 /* Cutslope production rate
/* equals cutslope production in tons divided by %YEARS%
additem %road_name%.aat cs_ttl_del_rate 15 15 n 3 /* Cutslope delivery rate
/* equals cutslope delivery in tons divided by %YEARS%

```

```

additem %road_name%.aat rd_sus_del_rate 15 15 n 3 /* Travelway suspended
/* sediment delivery rates
additem %road_name%.aat cs_sus_del_rate 15 15 n 3 /* Cutslope suspended
/* sediment delivery rates
additem %road_name%.aat un_ttl_del_rate 15 15 n 3 /* Travelway suspended
/* sediment delivery rates for ungraded roads
additem %road_name%.aat gr_ttl_del_rate 15 15 n 3 /* Travelway suspended
/* sediment delivery rates for graded roads
additem %road_name%.aat ab_ttl_del_rate 15 15 n 3 /* Travelway suspended
/* sediment delivery rates for abandoned roads
select %road_name%.aat
&type /&Beginning to calculate road erosion rates...
aselect
  reselect basin cn [quote %basin%]
  reselect grading = 'ungraded'
  calculate temp_sed = length_m * width_m * ( -0.432 + ( %ungraded% * %rain% * ( slope
  ** 1.5 ) ) ) ~
  * ( 1 + ( %silt_loss% * %silt_u_rd_fr% ) )
  reselect temp_sed < 0
  calculate temp_sed = 0
  aselect
    reselect basin cn [quote %basin%]
    reselect grading = 'graded'
    calculate temp_sed = length_m * width_m * ( -0.432 + ( %graded% * %rain% * ( slope **
    1.5 ) ) ) ~
    * ( 1 + ( %silt_loss% * %silt_g_rd_fr% ) )
    reselect temp_sed < 0
    calculate temp_sed = 0
    aselect
      reselect basin cn [quote %basin%]
      reselect grading = 'abandoned'
      calculate temp_sed = length_m * width_m * %rain% * %abandoned% * slope * ~
      ( 1 + ( %silt_loss% * %silt_a_rd_fr% ) )
      aselect
        reselect basin cn [quote %basin%]
        calculate cslope_kg = temp_sed * %cutslope% / 100
        calculate sed_kg = temp_sed - cslope_kg
        reselect sed_kg < 0
        calculate sed_kg = 0
        aselect
          calculate rd_ton = sed_kg / 1000
          calculate cs_ton = cslope_kg / 1000
          &type /&Beginning to calculate road sediment delivery rates for %watshed%...
          calculate rd_del = rd_ton * del_ratio
          calculate cs_del = cs_ton * del_ratio
          calculate rd_pr_rate = rd_ton / %years%
          calculate cs_pr_rate = cs_ton / %years%
          calculate rd_t_del_r = rd_del / %years%
          calculate cs_ttl_del_rate = cs_del / %years%
          reselect basin cn [quote %basin%]
          reselect grading = 'ungraded'
          calculate rd_sus_del_rate = rd_t_del_r * %un_sus_fr%
          calculate cs_sus_del_rate = cs_ttl_del_rate * %un_sus_fr%
          calculate un_ttl_del_rate = rd_t_del_r
          aselect
            reselect basin cn [quote %basin%]
            reselect grading = 'graded'
            calculate rd_sus_del_rate = rd_t_del_r * %gr_sus_fr%
            calculate cs_sus_del_rate = cs_ttl_del_rate * %gr_sus_fr%
            calculate gr_ttl_del_rate = rd_t_del_r
            aselect
              reselect basin cn [quote %basin%]
              reselect grading = 'abandoned'
              calculate rd_sus_del_rate = rd_t_del_r * %ab_sus_fr%
              calculate cs_sus_del_rate = cs_ttl_del_rate * %ab_sus_fr%
              calculate ab_ttl_del_rate = rd_t_del_r
              aselect
                &type /&Done calculating road erosion & delivery rates!
                dropitem %road_name%.aat (! gps_rds_ gps_rds_id area perimeter temp_sed ~
                sed_kg cslope_kg pt_del# pt_del-id del_ratio rd_ton rd_del cs_ton cs_del !)
                select %road_name%.aat

```

```

aselect
&type /&Unloading %road_name%.txt ...
reselect basin cn [quote %basin%]
unload %road_name%.txt length_m surface grading potential ~
  rd_pr_rate rd_t_del_r rd_sus_del_rate cs_pr_rate cs_ttl_del_rate ~
  cs_sus_del_rate delimited init
  /* creates a file with all of the results, replaces any file with same
  /* name, comma-delimited file may be opened in Excel
quit
kill pt_del all
build %road_name% line
&type /&Exiting rd_erosion routine...
&return

&ROUTINE STREAMBANK
/*****
/*
/* PURPOSE:      Calculates streambank sediment production and delivery rates for
/*               the basin chosen in the DEL_POTENTIAL routine using the
/*               attributes stored in the BANKS cover and the total time values
/*               chosen by the user during the SET_YEARS routine.
/*
/* REQUIRED       %BANK_ER% is the rate of sediment production from erodible
/* PRE-DEFINED   streambanks.
/* VARIABLES:    %YEARS% are the years over which sediment production
/*               and delivery rates are calculated.
/*
/* IMPORTANT     None
/* CREATED
/* VARIABLES:
/*
/* SIDE          (1)A new polygon cover called BANK_DEL is created.
/* EFFECTS       (2)TABLES is invoked.
/*
/* ALGORITHM:    The BANKS polygon cover is intersected with the DEL_BD cover created*
/*               during the DEL_POTENTIAL routine to assign DEL_RATIO values to each *
/*               stream segment according to its geographical location within
/*               %BASIN%. TABLES is invoked once again to add and calculate the
/*               value of items that will contain the sediment production and
/*               delivery estimates for each stream segment according to the
/*               values stored in the following items: PERIMETER, BANK_HT_M, and
/*               DEL_RATIO. The variables %BANK_ER% and %YEARS% are also used to
/*               calculate these items. Sediment delivery rates are estimated for
/*               both total loads and for the fraction of sediment finer than 2 mm,
/*               which is expected to be transported in suspension. Command is
/*               transferred back to the ARC prompt before the end of the routine.
/*
/*
*****/

&type /&Beginning streambank erosion routine...
intersect banks del_bd bank_del poly /*Used to assign delivery potential to
  /* banks cover
tables
additem bank_del.pat bank_ton 8 8 n 3 /* Bank erosion in tons
additem bank_del.pat bank_ttl_del 12 12 n 3 /* Bank delivery total in tons
additem bank_del.pat bank_sus_del 12 12 n 3 /* Bank suspended sediment
/* delivery total in tons
select bank_del.pat
reselect inside = 100 /* Used to avoid calculating erosion for universal polygon.
calculate bank_ton = ( perimeter / 2 ) * 2 * bank_ht_m * %years% ~
  * %bank_er% / 1000
calculate bank_ttl_del = bank_ton * del_ratio
calculate bank_sus_del = bank_ttl_del * %bank_sus_fr%
quit
&type /& Exiting streambank erosion routine...
&return

```

```

&ROUTINE STREAM_TOTAL
/*****
/*
/* PURPOSE:      Calculates treethrow sediment production and delivery rates for
/*               the basin chosen in the DEL_POTENTIAL routine using the
/*               STJ_STR line layer and the total time value chosen by the user
/*               during the SET_YEARS routine. The final polygon layer contains
/*               sediment production and delivery from both treethrow rates and
/*               streambank erosion.
/*
/*
/* REQUIRED      %TREETHROW% is the rate of sediment production by overthrown trees
/* PRE-DEFINED  in the proximity of streams.
/* VARIABLES:   %YEARS% is the total time in years over which sediment production
/*               and delivery rates are calculated.
/*
/*
/* IMPORTANT    None
/* CREATED
/* VARIABLES:
/*
/* SIDE        (1)A new polygon layer called STR_BD is created and killed.
/* EFFECTS      (2)A new polygon layer called BUF_STR is created and killed.
/*               (3)A new polygon layer called TR_DEL is created and killed.
/*               (4)A new polygon layer called TEMP_TRE is created and killed.
/*               (5)A new polygon layer called STR_DEL is created.
/*               (6)TABLES is invoked.
/*               (7)The polygon layer BANK_DEL is killed
/*
/*
/* ALGORITHM:   The STJ_STR line layer is clipped with DEL_BD-created
/*               during the DEL_POTENTIAL routine to eliminate stream segments
/*               located outside of %BASIN% in the new layer STR_BD.
/*               STR_BD is buffered to create the polygon layer BUF_STR. BUF_STR
/*               is then used in an UNION overlay with BANK_DEL-created
/*               during the STREAMBANK routine to assign bank erosion values to
/*               the stream polygons in BUF_STR. Some items in the new polygon layer
/*               (TR_DEL) are removed from its INFO table in TABLES in order to
/*               prepare it for the INTERSECT with DEL_BD which will
/*               assign delivery ratios to the polygons. Some items from this new
/*               polygon layer TEMP_TRE are dropped in TABLES to prepare it for the
/*               DISSOLVE command. DISSOLVE is performed using all remaining items
/*               so that polygons in original BANKS layer remain the same in
/*               order to avoid double counting of bank erosion. TABLES is invoked
/*               once again to add necessary items where the treethrow sediment
/*               production and delivery rate estimates are going to be stored.
/*               Treethrow production and delivery rates are a function of the
/*               PERIMETER and DEL_RATIO items and the %TREETHROW% and %YEARS%
/*               variables. Sediment delivery rates are estimated for both total
/*               loads and the fraction of sediment finer than 2mm, which is
/*               expected to be transported in suspension. The TABLES session is
/*               then terminated and the STR_BD, BUF_STR, TEMP_TRE, TR_DEL, and
/*               BANK_DEL layers are killed before the end of the routine.
/*
/*
*****/

&type /&Beginning with stream_total routine...
clip stj_str del_bd str_bd line
buffer str_bd buf_str # 0.25 # line
union buf_str bank_del tr_del
tables
dropitem tr_del.pat (! potential del_ratio !) /* Done to avoid having empty
/* records in temp_tre after intersect command
quit
intersect tr_del del_bd temp_tre poly
tables
dropitem temp_tre.pat (! del_bd-id del_bd# tr_del# tr_del-id buf_str# buf_str-id ~
inside bank_del# bank_del-id banks# banks_banks_id !) /* Needed for dissolve
/* function to work as intended- to dissolve along all remaining items.
select temp_tre.pat
reselect bank_ht_m = 0 /* Selects polygons that were not in the BANKS cover.
/* These records actually have no value in the bank_ht_m item, but can be
/* selected with this command.
reselect temp_tre-id > 0 /* Eliminates the Universal polygon from the selection.

```

```

calculate bank_ht_m = 0 /* Assigns a 0 value to polygons not included in BANKS.
/* This step is also needed for the dissolve command to dissolve along polygons
/* in the BANKS original coverage and avoid double counting of bank erosion.
quit
dissolve temp_tre str_del #all
tables
additem str_del.pat tree_ton 8 8 n 3
additem str_del.pat tree_ttl_del 12 12 n 3
additem str_del.pat tree_sus_del 12 12 n 3
select str_del.pat
reselect potential = ''
nselect
calculate tree_ton = ( perimeter / ( 2 * 1000 ) ) * %years% * %treethrow%
calculate tree_ttl_del = tree_ton * del_ratio
calculate tree_sus_del = tree_ttl_del * %tree_sus_fr%
quit
kill (! str_bd buf_str temp_tre tr_del bank_del !) all
&type /&Exiting stream_total routine.../&
&return

&ROUTINE SURF_EROSION
/*****
/*
/* PURPOSE:      Calculates sediment production and delivery rates by surface
/*               erosion on undisturbed hillslopes for the basin chosen in the
/*               DEL_POTENTIAL routine.
/*
/*
/* REQUIRED      %UNDISTURBED% is the rate of sediment production by surface
/* PRE-DEFINED  erosion on undisturbed hillslopes.
/* VARIABLES:   %YEARS% is the total time in years over which sediment production
/*               and delivery rates are calculated.
/*               %RAIN_RATE% is the annual rainfall rate in cm per year.
/*               %RAIN% is the total rainfall in centimeters.
/*               %SILT_LOSS% is the ratio of actual sediment production for silt-
/*               sized material to that measured with sediment traps.
/*
/* IMPORTANT    None
/* CREATED
/* VARIABLES:
/*
/* SIDE        (1)A new line layer called TEMP_BD is created and killed.
/* EFFECTS     (2)A new polygon layer called BUF_CST is created and killed.
/*             (3)A new polygon layer called CST_BD is created and killed.
/*             (4)A new polygon layer called SE_BD is created.
/*             (5)TABLES is invoked.
/*
/* ALGORITHM:  A copy of STJ_BD called TEMP_BD, which contains the entire St. John*
/* coastline is built as a line. A 30 m buffer is created around it
/* and stored as the BUF_CST layer. This layer is then clipped by
/* %BASIN%_BD to eliminate the entire buffered coastline with the
/* excetion of the landward side of the coastline inside %BASIN%. The
/* resulting layer (CST_BD)is then used in an UNION command with
/* DEL_BD to assign the DEL_RATIO to the new layer SE_BD. TABLES is
/* invoked to add items where production and delivery data will
/* be stored. Manipulations using the reselect command avoid
/* calculating sediment production and delivery for wetland areas
/* and areas within 30 m of the coastline. Production and delivery
/* are calculated as a function of the AREA and DEL_RATIO items and
/* the %UNDISTURBED% and %RAIN% variables. Sediment delivery ratios
/* are estimated for both total loads and the fraction finer than
/* 2 mm, which is expected to be transported in suspension. The
/* TABLES session is then terminated and the TEMP_BD, BUF_CST, and
/* CST_BD layers are killed before the end of the routine.
/*
/* *****/
&type /& Beginning with surface erosion routine.../&
copy stj_bd temp_bd
build temp_bd line
buffer temp_bd buf_cst # 30 # line /* Creates a buffer around entire coastline.

```



```

/* No surface erosion is calculated for buffered coastline area.
clip buf_cst %basin%_bd cst_bd poly /* Conserves buffer for %basin% only
union cst_bd del_bd se_bd /* Combines buffered coastline with del_bd.
/* This then allows to calculate sed prod and delivery for each area and
/* to prevent calculating erosion for wetland and coastline areas.
tables
additem se_bd.pat se_kg 8 8 n 2
additem se_bd.pat se_ton 8 8 n 3
additem se_bd.pat se_ttl_del 12 12 n 3
additem se_bd.pat se_sus_del 12 12 n 3
select SE_BD.PAT
&type /* Calculating surface erosion on undisturbed surfaces.../&
reselect potential = '' /* Selects Universal polygon
nselect /* Eliminates universal polygon.
reselect inside = 1 /* Eliminates buffered coastline areas.
calculate se_kg = ( area * 0.14 * %rain% * %undisturbed% ) * ~
( 1 + ( %silt_loss% * %silt_se_fr% ) ) /* This last portion of the equation is
/* meant to compensate for loss of silt over sediment traps that were
/* used to develop the empirical erosion rate function.
aselect
reselect potential = 'wetland'
calculate se_kg = 0 /* Surface erosion is not calculated for wetland areas.
aselect
calculate se_ton = se_kg / 1000
calculate se_ttl_del = se_ton * del_ratio
calculate se_sus_del = se_ttl_del * %se_sus_fr%
quit
kill (! temp_bd buf_cst cst_bd !) all
&type /*Exiting surface erosion routine.../&
&return

&ROUTINE NAT_EROSION
/*****
/*
/* PURPOSE: Places all sediment production and delivery estimates derived from*/
/* natural processes in a single polygon layer and text file. */
/*
/* REQUIRED %NAT_NAME% is the user-given name of the final layer and text */
/* PRE-DEFINED file containing the final results of the natural erosion and */
/* VARIABLES: sediment delivery analysis. */
/* %YEARS% is the total time in years over which sediment production */
/* will be calculated. */
/*
/* IMPORTANT None */
/* CREATED */
/* VARIABLES: */
/*
/* SIDE (1)A new polygon cover called %NAT_NAME% is created. */
/* EFFECTS (2)TABLES is invoked. */
/*
/* ALGORITHM: An UNION command is used to join STR_DEL and SE_BD into one single */
/* layer called %NAT_NAME% that contains sediment production and */
/* delivery estimates from treethrow, streambank erosion, and surface */
/* erosion of undisturbed hillslopes. TABLES is invoked and several */
/* manipulations are used to avoid double counting of surface erosion */
/* rates. Items that contain total sediment production and delivery */
/* estimates from natural processes are added and calculated. */
/* Production and delivery are calculated as a function of the */
/* SE_TON,BANK_TON, TREE_TON, SE_DEL, BANK_DEL, and TREE_DEL items */
/* and the %YEARS% variable. A file named %NAT_NAME%.txt containing */
/* all of the results is unloaded before the TABLES session is */
/* terminated. The SE_BD and STR_DEL layers are killed before exiting */
/* the routine. */
/*
/*
*****/

&type /*Beginning nat_erosion routine...
union se_bd str_del %nat_name% /* Unions treethrow, streambank and surface erosion
/* estimates into one coverage.
tables
select %nat_name%.pat

```

```

reselect tree_ton > 0
calculate se_ton = 0 /* Avoids double counting of surface erosion estimates.
calculate se_ttl_del = 0 /* Avoids double counting of sediment delivery estimates.
calculate se_sus_del = 0
aselect
additem %nat_name%.pat nat_ton 8 8 n 2
additem %nat_name%.pat nat_ttl_del 12 12 n 2 /* Total sediment delivery
additem %nat_name%.pat nat_sus_del 12 12 n 2 /* Suspended sediment delivery
additem %nat_name%.pat nat_pr_rate 12 12 n 2 /* Total sediment production
additem %nat_name%.pat nat_t_del_rate 12 12 n 2 /* Total delivery rate
additem %nat_name%.pat nat_s_del_rate 12 12 n 2 /* Suspended delivery rate
select %nat_name%.PAT
calculate nat_ton = se_ton + bank_ton + tree_ton
calculate nat_pr_rate = nat_ton / %years%
calculate nat_ttl_del = bank_ttl_del + tree_ttl_del + se_ttl_del
calculate nat_sus_del = bank_sus_del + tree_sus_del + se_sus_del
calculate nat_t_del_rate = nat_ttl_del / %years%
calculate nat_s_del_rate = nat_sus_del / %years%
dropitem %nat_name%.pat (! se_bd# se_bd-id cst_bd# cst_bd-id inside del_bd# ~
del_bd-id se_kg str_del# str_del-id !)
&type Unloading %nat_name%.txt ...
unload %nat_name%.txt area potential se_ton se_ttl_del se_sus_del bank_ton ~
bank_ttl_del bank_sus_del tree_ton tree_ttl_del tree_sus_del nat_pr_rate ~
nat_t_del_rate nat_s_del_rate delimited init
/* creates a file with all of the results,
/* replaces any file with same name, comma-delimited file may be opened
/* from Microsoft Excel
quit
kill (! se_bd str_del !) all
build %nat_name% poly
&type /&Done with nat_erosion routine...
&return

&ROUTINE SUMMARY_RESULTS
/*****
/*
/* PURPOSE:      Displays a summary of results for both road and natural erosion
/*               and delivery rates on the screen.
/*
/*
/* REQUIRED      %NAT_NAME% is the user-given name of the final cover and text
/* PRE-DEFINED  file containing the final results of the natural erosion and
/* VARIABLES:   sediment delivery analysis.
/*
/*             %ROAD_NAME% is the user-given name of the final cover and text
/*             file containing the final results of the road erosion and
/*             sediment delivery analysis.
/*
/*
/* IMPORTANT    A total of 22 new variables are created by this routine.
/* CREATED
/* VARIABLES:
/*
/* SIDE         (1) ARCEDITS invoked.
/* EFFECTS      (2) TABLES is invoked.
/*
/*
/* ALGORITHM:   The statistics command is used to sum the sediment delivery totals
/*               of a number of items in %NAT_NAME%.PAT and %ROAD_NAME%.AAT.
/*               ARCEDIT is invoked to set the values of several new variables.
/*               After exiting ARCEDIT the TABLES module is invoked to drop
/*               a number of items from the %ROAD_NAME% data layer. Finally, the
/*               routine prints on the screen a table summarizing the overall
/*               model results.
/*
/*
/*****
&if [exists %nat_name%.sta] &then kill %nat_name%.sta
&if [exists %road_name%.sta] &then kill %road_name%.sta
statistics %nat_name%.pat %nat_name%.sta
sum bank_ttl_del
sum bank_sus_del
sum tree_ttl_del
sum tree_sus_del
sum se_ttl_del
sum se_sus_del

```

```

sum nat_ttl_del
sum nat_sus_del
end
statistics %road_name%.aat %road_name%.sta
sum un_ttl_del_rate
sum gr_ttl_del_rate
sum ab_ttl_del_rate
sum cs_ttl_del_rate
sum cs_sus_del_rate
end
arccedit
edit %nat_name%.sta info
&sv bank_ttl_del_r = [show info 1 item sum-bank_ttl_del] / %years%
&sv bank_sus_del_r = [show info 1 item sum-bank_sus_del] / %years%
&sv tree_ttl_del_r = [show info 1 item sum-tree_ttl_del] / %years%
&sv tree_sus_del_r = [show info 1 item sum-tree_sus_del] / %years%
&sv se_ttl_del_r = [show info 1 item sum-se_ttl_del] / %years%
&sv se_sus_del_r = [show info 1 item sum-se_sus_del] / %years%
&sv nat_ttl_del_r = [show info 1 item sum-nat_ttl_del] / %years%
&sv nat_sus_del_r = [show info 1 item sum-nat_sus_del] / %years%
edit %road_name%.sta info
&sv un_ttl_del_r = [show info 1 item sum-un_ttl_del_rate]
&sv gr_ttl_del_r = [show info 1 item sum-gr_ttl_del_rate]
&sv ab_ttl_del_r = [show info 1 item sum-ab_ttl_del_rate]
&sv un_sus_del_r = %un_ttl_del_r% * %un_sus_fr%
&sv gr_sus_del_r = %gr_ttl_del_r% * %gr_sus_fr%
&sv ab_sus_del_r = %ab_ttl_del_r% * %ab_sus_fr%
&sv cs_ttl_del_r = [show info 1 item sum-cs_ttl_del_rate]
&sv cs_sus_del_r = [show info 1 item sum-cs_sus_del_rate]
&sv rd_ttl_del_r = ( %un_ttl_del_r% + %gr_ttl_del_r% + %ab_ttl_del_r% + %cs_ttl_del_r%
)
&sv rd_sus_del_r = ( %un_sus_del_r% + %gr_sus_del_r% + %ab_sus_del_r% + %cs_sus_del_r%
)
&sv min_cur_nat = ( ( %rd_sus_del_r% + %nat_ttl_del_r% ) / %nat_ttl_del_r% )
&sv max_cur_nat = ( ( %rd_ttl_del_r% + %nat_sus_del_r% ) / %nat_sus_del_r% )
&sv total_sus_del_r = ( %bank_sus_del_r% + %tree_sus_del_r% + %se_sus_del_r% + ~
%un_sus_del_r% + %gr_sus_del_r% + %ab_sus_del_r% + %cs_sus_del_r% )
&sv total_del_r = ( %bank_ttl_del_r% + %tree_ttl_del_r% + %se_ttl_del_r% + ~
%un_ttl_del_r% + %gr_ttl_del_r% + %ab_ttl_del_r% + %cs_ttl_del_r% )
quit
tables
dropitem %road_name%.aat (! un_ttl_del_rate gr_ttl_del_rate ab_ttl_del_rate ~
rd_sus_del_rate cs_sus_del_rate rd_pr_rate cs_pr_rate !)
quit
&type /&/&
&type _____/&
&type Source name      Susp. del. rate      Total del. rate
&type                (tons/year)        (tons/year)/&
&type _____/&
&type Streambank      [round %bank_sus_del_r% ]      [round %bank_ttl_del_r% ]
&type Treethrow        [ %tree_sus_del_r% ]          [ %tree_ttl_del_r% ]
&type Undisturbed      [ %se_sus_del_r% ]            [ %se_ttl_del_r% ]
&type Ungraded rds     [round %un_sus_del_r% ]        [round %un_ttl_del_r% ]
&type Graded rds       [round %gr_sus_del_r% ]        [round %gr_ttl_del_r% ]
&type Abandoned rds   [ %ab_sus_del_r% ]            [ %ab_ttl_del_r% ]
&type Cutslopes       [round %cs_sus_del_r% ]        [round %cs_ttl_del_r% ]
/&
&type TOTAL           [ round %total_sus_del_r% ]      [round %total_del_r% ]
&type _____/&/&
&type Natural sediment yield rates for the %watshed% basin range from
&type      [round %nat_sus_del_r% ] to [round %nat_ttl_del_r% ] tons per year./&
&type Road sediment yield rates for the %watshed% basin range from
&type      [round %rd_sus_del_r% ] to [round %rd_ttl_del_r% ] tons per year./&
&type Current sediment delivery rates into %watshed% range between
&type      [round %min_cur_nat% ] and [round %max_cur_nat% ] times above undisturbed
conditions.
&type /&/&
&if [query 'Press enter to exit.' .true.] &then ~
&return

```